

## Задания

### Задание 27 № 15643

На вход программы поступает последовательность из  $N$  целых положительных чисел, все числа в последовательности различны. Рассматриваются все пары различных элементов последовательности (элементы пары не обязаны стоять в последовательности рядом, порядок элементов в паре не важен). Необходимо определить количество пар, для которых произведение элементов не делится на 34.

#### Описание входных и выходных данных

В первой строке входных данных задаётся количество чисел  $N$  ( $1 \leq N \leq 1000$ ). В каждой из последующих  $N$  строк записано одно целое положительное число, не превышающее 10 000. В качестве результата программа должна напечатать одно число: количество пар, в которых произведение элементов не кратно 34.

*Пример входных данных:*

5  
3  
4  
10  
11  
17

*Пример выходных данных для приведённого выше примера входных данных:*

8

*Пояснение.* Из пяти заданных чисел можно составить 10 попарных произведений:  $3 \cdot 4$ ,  $3 \cdot 10$ ,  $3 \cdot 11$ ,  $3 \cdot 17$ ,  $4 \cdot 10$ ,  $4 \cdot 11$ ,  $4 \cdot 17$ ,  $10 \cdot 11$ ,  $10 \cdot 17$ ,  $11 \cdot 17$  (результаты: 12, 30, 33, 51, 40, 44, 68, 110, 170, 187). Из них на 34 не делятся 8 произведений ( $3 \cdot 4=12$ ,  $3 \cdot 10=30$ ,  $3 \cdot 11=33$ ,  $3 \cdot 17=51$ ,  $4 \cdot 10=40$ ,  $4 \cdot 11=44$ ,  $10 \cdot 11=110$ ,  $11 \cdot 17=187$ ).

Требуется написать эффективную по времени и по памяти программу для решения описанной задачи.

Программа считается эффективной по времени, если при увеличении количества исходных чисел  $N$  в  $k$  раз время работы программы увеличивается не более чем в  $k$  раз.

Программа считается эффективной по памяти, если память, необходимая для хранения всех переменных программы, не превышает 1 Кбайт и не увеличивается с ростом  $N$ .

Максимальная оценка за правильную (не содержащую синтаксических ошибок и дающую правильный ответ при любых допустимых входных данных) программу, эффективную по времени и по памяти, — 4 балла.

Максимальная оценка за правильную программу, эффективную только по времени — 3 балла.

Максимальная оценка за правильную программу, не удовлетворяющую требованиям эффективности, — 2 балла.

Вы можете сдать одну программу или две программы решения задачи (например, одна из программ может быть менее эффективна). Если Вы сдадите две программы, то каждая из них будет оцениваться независимо от другой, итоговой станет большая из двух оценок.

Перед текстом программы обязательно кратко опишите алгоритм решения. Укажите использованный язык программирования и его версию.

#### Решение.

Решение эффективное по памяти и времени (Задача Б). Прямой вариант.

Решение основано на подсчете количества чисел кратных 2 и не кратных 17, кратных 17 и не кратных 2, не кратных 2 и не кратных 17. После чего, найдя количество комбинаций чисел из этих групп, получается ответ.

Обозначим за  $k_2$  – количество чисел кратных 2 и не кратных 17, за  $k_{17}$  – количество чисел кратных 17 и не кратных 2, за  $k$  – количество чисел не кратных 2 и не кратных 17. Тогда количество чисел будет равно

$$k_2 \cdot (k_2 - 1) / 2 + k_{17} \cdot (k_{17} - 1) / 2 + k \cdot (k - 1) / 2 + k_2 \cdot k + k_{17} \cdot k$$

Программа на языке Pascal.

```
var k2, k17, k: integer;
    n, i, x: integer;
```

```

begin
  k := 0; k2 := 0; k17 := 0;
  readln(n);
  for i:=1 to n do begin
    readln(x);
    if x mod 34 <> 0 then begin
      if x mod 2 = 0 then k2 := k2 + 1
      else if x mod 17 = 0 then k17 := k17 + 1
      else k := k + 1;
    end;
  end;
  writeln((k2*(k2-1)+k17*(k17-1)+k*(k-1))/2+k2*k+k17*k);
end.

```

Решение эффективное по памяти и времени (Задача Б). Решение от обратного.

Решение основано на подсчете количества чисел кратных 2 и не кратных 17, кратных 17 и не кратных 2, кратных 34. После чего, найдя количество комбинаций чисел из этих групп, получается ответ.

Обозначим за  $k_2$  – количество чисел кратных 2 и не кратных 17, за  $k_{17}$  – количество чисел кратных 17 и не кратных 2, за  $k_{34}$  – количество чисел кратных 34,  $n$  – количество введенных чисел.

Тогда количество чисел будет равно общему количеству за вычетом произведения количества чисел кратных 34 на общее количество и произведения количества чисел, кратных 2, и количества чисел кратных 17.

Программа на языке Pascal.

```

var k2, k17, k34: integer;
    n, i, x: integer;
begin
  k34 := 0; k2 := 0; k17 := 0;
  readln(n);
  for i:=1 to n do begin
    readln(x);
    if x mod 34 = 0 then
      k34 := k34 + 1
    else if x mod 2 = 0 then
      k2 := k2 + 1
    else if x mod 17 = 0 then
      k17 := k17 + 1;
  end;

  writeln(n*(n-1)/2-k34*(n-k34)-k34*(k34-1)/2-k2*k17);
end.

```

Решение не эффективное по памяти и по времени (Задача А).

```

var N: integer;
    a: array[1..10000] of integer;
    i, j, k: integer;
begin
  readln(N);
  for i:=1 to N do read(a[i]);
  k:= 0;
  for i:= 1 to N-1 do
    for j:= i+1 to N do
      if a[i]*a[j] mod 34 <> 0 then

```

```
        k := k + 1;  
writeln(k)  
end.
```